

OFFICE DE LA RECHERCHE SCIENTIFIQUE ET TECHNIQUE OUTRE-MER
MISSION ORSTOM DE PORT-VILA

NOTES ET DOCUMENTS D'OCEANOGRAPHIE

N° 10

SEPTEMBRE 1984

CROISSANCE DES JEUNES THONS JAUNES (THUNNUS ALBACARES)
ET DES BONITES (KATSUWONUS PELAMIS)
DANS LE PACIFIQUE TROPICAL OCCIDENTAL

BROUARD, F.

GRANDPERRIN, R.

CILLAURREN, E.

MISSION ORSTOM

B.P. 76 - PORT-VILA

VANUATU

20 NOV. 1984

O. R. S. I. O. M. Fonds Documentaire

N° : 16 465, ex 3

Cote A

Croissance des jeunes thons jaunes (Thunnus albacares)
et des bonites (Katsuwonus pelamis)
dans le Pacifique tropical occidental

Table des matières

	page
RESUME.....	2
SUMMARY.....	2
INTRODUCTION.....	3
MATERIEL ET METHODES.....	3
Pêche, collecte et saisie des données.....	3
Analyse de la croissance.....	5
RESULTATS ET DISCUSSIONS.....	6
Croissance de <u>Thunnus albacares</u>	8
Croissance de <u>Katsuwonus pelamis</u>	9
CONCLUSION.....	13
REMERCIEMENTS.....	13
BIBLIOGRAPHIE.....	14
ANNEXE 1 - Logiciel d'analyse des distributions de fréquences de tailles dérivé du programme ELEFAN 1 de PAULY et DAVID (1981).....	16
ANNEXE 2 - Restructuration des échantillons (<u>Thunnus albacares</u> , Vanuatu).....	22

R E S U M E

Ce document décrit la croissance de jeunes thons jaunes (Thunnus albacares) et de bonites (Katsuwonus pelamis) capturés à la traîne en 1983 autour de dispositifs de concentration de poissons (DCP) situés dans le sud-ouest de l'île d'Efaté, à Vanuatu. La méthode utilisée est basée sur le suivi de la progression des modes des distributions de fréquences de tailles. L'analyse des données a été réalisée au moyen d'un programme écrit en BASIC qui est une modification du programme ELEFAN 1 de PAULY et DAVID (1981). La modification consiste à identifier plusieurs valeurs de L_{∞} et k correspondant à des lignes de crêtes d'ESP ("Explained Sum of Peaks"). Pour des thons jaunes de tailles comprises entre 30 et 50 cm, la croissance moyenne serait de 1,3 cm par mois ; elle augmenterait par la suite. La croissance moyenne des bonites comprises entre 30 et 50 cm serait de 1,9 cm par mois. La représentation des ESP et les courbes de croissance figurent dans le texte. Le logiciel, ainsi qu'un exemple de restructuration des données, sont reproduits en annexes.

S U M M A R Y

This paper describes the growth of small yellowfin tuna (Thunnus albacares) and skipjack (Katsuwonus pelamis) trolled in 1983 around fish aggregating devices (FADs) deployed south west of Efate Island in Vanuatu. The growth parameters have been calculated from length - frequency data using a BASIC program modified from ELEFAN 1 (PAULY and DAVID, 1981). This program allows the calculation of several values of L_{∞} and k which are represented by several "Explained Sums of Peaks" (ESP). The average growth rate of yellowfin in the size range 30 - 50 cm was 1.3 cm per month ; it might increase with age. The growth rate of skipjack between 30 and 50 cm averaged 1.9 cm per month. Diagrams showing ESP and growth curves are included. The BASIC program as well as an example of data transformation are given as appendices.

Croissance des jeunes thons jaunes (Thunnus albacares)
et des bonites (Katsuwonus pelamis)
dans le Pacifique tropical occidental

I N T R O D U C T I O N

Au cours de l'année 1982, trois dispositifs de concentration de poissons (DCP) ont été mouillés au large de l'île Efate à Vanuatu (fig.1). Les données collectées au cours de pêches effectuées autour de ces radeaux (GRANDPERRIN et BROUARD, 1983) ont permis d'étudier la croissance des deux principales espèces : thons à nageoires jaunes (Thunnus albacares) et bonite à ventre rayé (Katsuwonus pelamis).

M A T E R I E L E T M E T H O D E S

Pêche, collecte et saisie des données

Les pêches furent réalisées à la traîne de surface par des embarcations de 8,5 m type catamaran Alia. Bien que les trois DCP aient été fréquentés de façon régulière, l'essentiel des prises de l'année 1983 provenait toutefois de DCP 2. Au débarquement, les poissons étaient mesurés, de telle sorte que les distributions de fréquences ont pu être établies mois après mois ; les gonades étaient examinées puis pesées pour identification des sexes et des stades de maturation ; les premiers rayons de chacune des deux nageoires dorsales étaient prélevés dans le but de réaliser une étude ultérieure sur l'ageage par examen des stries journalières d'accroissement. Les calculs ont été réalisés sur un microordinateur HP 85.




Fig. 1 - Dispositifs de concentration de poissons à Efaté.

Analyse de la croissance

Il est généralement admis que le modèle de Von BERTALANFFY $L = L_{\infty}(1 - e^{-k(t-t_0)})$, rend bien compte de la croissance des poissons. Le problème consiste à déterminer les paramètres L_{∞} , k et t_0 qui ajusteront au mieux le modèle aux observations. En règle générale, les valeurs trouvées ne seront applicables qu'à l'intérieur d'un intervalle de tailles donné, les extrapolations en dehors des limites étant le plus souvent hasardeuses (JOSSE et al., 1979 ; LAUREC et LE GUEN, 1981). Dans la présente étude, l'analyse des distributions de fréquences de longueurs des thons jaunes ayant été faite sur des poissons inférieurs à 60 cm, il serait illusoire de vouloir utiliser les paramètres que nous allons calculés pour prévoir la croissance au-delà de cette taille.

Parmi les différentes méthodes d'ajustement disponibles, nous avons retenu celle de PAULY (PAULY et DAVID, 1981 ; PAULY et al., 1980) en la modifiant toutefois de façon substantielle. Cette méthode, basée sur l'utilisation du programme ELEFAN I écrit en BASIC consiste à maximaliser un critère appelé ESP ("Explained Sum of Peaks") qui est calculé comme suit. Soit l'ensemble des distributions de fréquences correspondant chacune à un instant T . On fait subir à cet ensemble des translations dans le temps autant de fois que le nécessite la longévité apparente de l'espèce. Etant donné un couple de paramètres (L_{∞} , k), on calcule pour chaque T la longueur théorique des poissons. A cette longueur correspond une fréquence observée. L'ESP est, dans son principe, la somme de ces fréquences. Selon PAULY et DAVID (1981) le meilleur couple (L_{∞} , k) sera celui qui fournira l'ESP maximum. En d'autres termes, la courbe de croissance ainsi définie sera celle qui passera par le plus grand nombre de "pics" de fréquences de longueurs et par le minimum de "creux" (1).

Cette méthode présente trois inconvénients :

(a) Ne fixant pas les limites de recherche de la solution optimale, elle conduit parfois, après de laborieux calculs par itérations successives, à des valeurs de L_{∞} et k correspondant à des cas absurdes.

(1) Il s'agit des distributions de fréquences restructurées (Annexes 1 et 2).

(b) Elle ne fournit qu'un seul ESP maximum, c'est-à-dire un seul couple de valeurs (L_{00} , k) alors qu'il existe tout un ensemble de maximums locaux, chacun d'eux présentant à priori le même intérêt que les autres.

(c) Le critère de l'ESP maximum n'a pas de sens biologique en soi. Il correspond à certaines hypothèses arbitraires sur la validité des modes. Elles peuvent ne pas correspondre à la réalité. En particulier, elles impliquent que les modes surnuméraires ainsi que l'absence de modes soient improbables. Elles conduisent aussi à l'affirmation qu'entre deux couples (L_{00} , k) dont l'un suggère qu'un mode apparent est en fait la confusion de deux et l'autre qu'il n'en existe qu'un, il faille toujours choisir le premier.

Afin de remédier à ces inconvénients, nous proposons quelques modifications au programme ELEFAN I (Annexe 1). La première consiste à ne calculer les ESP qu'entre certaines limites de L_{00} et k . La seconde fait intervenir un espace à trois dimensions défini par les axes des k , des L_{00} et des ESP. L'ensemble des valeurs d'ESP est assimilable aux cotes d'une surface dont la représentation graphique se fait par projection sur le plan défini par les axes des L_{00} et k , les crêtes (ESP maximales) correspondant aux zones foncées et les creux (ESP minimales) aux zones claires (fig. 2). On obtient ainsi une vision synthétique de l'ensemble des couples (L_{00} , k) possibles, les plus plausibles étant associés aux crêtes. Des informations exogènes aux distributions de fréquences permettront alors de choisir le meilleur couple (L_{00} , k). A ce niveau intervient indiscutablement une grande part de subjectivité - mais c'est inéluctable - en partie atténuée par l'expérience.

R E S U L T A T S E T D I S C U S S I O N S

Les individus capturés au large la façade sud-ouest d'Efaté appartiennent, soit à des stocks sédentaires, soit, et c'est plus probable, à des groupes en migration. Les études de croissance basées sur le suivi de la progression des modes n'ont donc de sens que si on formule l'hypothèse selon laquelle les échantillons sont représentatifs d'une population homo-

SURFACE DES ESP
THUNNUS albacares




Fig. 2 - Thunnus albacares (Vanuatu) : surface des ESP




Fig. 3 - Thunnus albacares (Vanuatu) : courbes de croissance

gène occupant tout le S.W. Pacifique tropical.

Croissance de *Thunnus albacares*

Le graphique de la figure 2 fait apparaître une ligne de crêtes d'ESP. Elle correspond à $61 \text{ cm} < L < 67,5 \text{ cm}$ avec $0,55 < k < 0,7$. La courbe de croissance de *T. albacares* a été représentée sur la figure 3 pour trois couples de paramètres envisageables : (61 - 0,7), (60 - 0,6) et (67 - 0,6). Ils conduisent à des vitesses de croissance de l'ordre de 1 cm/mois pour des individus de taille comprise entre 30 cm et 50 cm. Il est intéressant de comparer ces résultats à ceux de la littérature (tableau 1). A l'exception de celles trouvées par FONTENEAU (1979), on constate que les vitesses de croissance fournies par les autres auteurs sont plus grandes que celles que nous avons calculées. Si les valeurs obtenues par MARCILLE et STEQUERT (1976) paraissent plausibles au-delà de 55 cm, il semble que le passage, de mai à décembre 1974, d'un mode de 40 cm à un mode de 50 cm, correspondrait plutôt à une croissance de 1,4 cm/mois qu'à 3 cm/mois. Quant aux résultats obtenus par LE GUEN et SAKAGAWA (1973) pour les thons jaunes de 30 à 50 cm, ils sont difficilement comparables aux nôtres dans la mesure où ils ont été obtenus par extrapolation en dehors des limites de distributions de fréquences de tailles qu'ils ont observées.

Tableau 1 : Estimations de la croissance des thons jaunes

	Croissance (cm/mois)	Références
Atlantique Est (1)	5,3	LE GUEN et SAKAGAWA (1973)
Atlantique Est (2)	2,8	LE GUEN et SAKAGAWA (1973)
Atlantique Est (1)	1,4	FONTENEAU (1979)
Atlantique Est (2)	3	FONTENEAU (1979)
Océan Indien Ouest (1)	3	MARCILLE et STEQUERT (1976)
Pacifique Central (3)	3,7	UCHIYAMA et STRUHSAKER (1981)
Pacifique Ouest (4)	1,3	Présente étude

(1) $40 \text{ cm} < L < 70 \text{ cm}$ fréquences

(2) $70 \text{ cm} < L < 140 \text{ cm}$ fréquences

(3) $30 \text{ cm} < L < 50 \text{ cm}$ lecture d'otolithes

(4) $30 \text{ cm} < L < 50 \text{ cm}$ fréquences

Pour expliquer l'évolution des modes observés à Vanuatu, qui traduirait une croissance relativement faible, deux hypothèses peuvent être formulées. Dans la première, on rejette le fait qu'ils soient attribuables à la croissance. On peut ainsi supposer, dans le cas d'un recrutement continu, que l'apparition des individus d'un âge donné, donc d'une taille bien définie, soit liée à certains paramètres du milieu (température, salinité, etc...) ; la progression modale traduirait alors une évolution saisonnière des conditions de l'environnement. En fait, il semble que le recrutement ne soit pas continu, des périodes de repos alternant avec des périodes d'activité reproductrice intense (ALBARET, 1977). Il se peut aussi que la sélectivité que les DCP semblent exercer modifie les distributions de fréquences de tailles. Certains auteurs admettent en effet que les gros individus s'éloignent du radeau durant la journée pour y revenir le soir (PRESTON, 1982 ; SHOMURA et MATSUMOTO, 1982). Au-delà d'une certaine taille, l'échantillon fourni par la traîne de surface ne serait donc pas représentatif de la population présente dans le milieu, et les valeurs modales apparentes ne seraient que des queues de distribution.

La seconde hypothèse consiste à attribuer des vitesses de croissance différentes aux jeunes et aux adultes. Selon FONTENEAU (1979) la plupart des informations disponibles conduisent à une telle conclusion. Il ajoute que différentes contraintes écophysiologiques telles que températures élevées et pauvreté des eaux peuvent expliquer ces différences. Pour BAYLIFF (1973) et SUZUKI (1971), ce fait est confirmé par les campagnes de marquage. Utilisant les données de CHABANNE et al (1982) recueillies sur les thons jaunes de Polynésie Française, nous avons retrouvé une croissance faible (fig. 4 et 5).

Croissance de *Katsuwonus pelamis*

Deux lignes de crêtes d'ESP sont observables sur la figure 6. La première correspond à $60 \text{ cm} < L_{\infty} < 64 \text{ cm}$ pour $0,6 < k < 0,8$ et la seconde à $60 \text{ cm} < L_{\infty} < 64 \text{ cm}$ pour $1,0 < k < 1,25$. Les courbes de croissance ont été tracées sur la figure 7 pour les deux couples de paramètres (60 - 0,75) et




Fig. 4 - Thunnus albacares (Tahiti) : surface des ESP





Fig. 5 - Thunnus albacares (Tahiti) : courbes de croissance

SURFACE DES ESP
KATSUWONUS pelamis



Echantillon n°12, 08
Longueur inf. 41.7
ESPM (n = 185) ESPmax 7.2003

Fig. 6 - Katsuwonus pelamis (Vanuatu) : surface des ESP




Fig. 7 - Katsuwonus pelamis (Vanuatu) : courbes de croissance

(62 - 1,1). Ces deux courbes correspondent à deux hypothèses différentes sur le recrutement. La première (60 - 0,75) implique un seul recrutement annuel. La courbe tend à passer par presque tous les modes ; le léger décalage observé en mai et juin entre la longueur théorique et la longueur modale peut être dû à la sélectivité des hameçons, le mode réel de la population in situ correspondant en fait à la classe de tailles par laquelle passe la courbe. L'accroissement mensuel moyen est de 1,1 cm. La deuxième courbe (62 - 1,1) implique deux recrutements annuels ; toutefois elle ne suit que la croissance des individus issus du second en délaissant ceux du premier (premiers modes en mai, juin et août, second mode en janvier). L'accroissement mensuel moyen est de 1,9 cm.

Le tableau 2 indique les paramètres de croissance obtenus par d'autres auteurs. On constate que les vitesses de croissance qu'ils calculent sont voisines de celle que fournit l'hypothèse 2 basée sur l'existence de deux recrutements. Comme les données de CHABANNE et al (1982) tendent aussi à montrer l'existence de recrutements différents et que STEQUERT (1976) observe plusieurs périodes de ponte durant la saison chaude en Océan Indien, nous avons retenu l'hypothèse 2. Elle correspond à une croissance moyenne égale à 1,9 cm par mois pour des bonites de taille comprise entre 30 et 50 cm.

Tableau 2 : Estimation de la croissance des bonites

	L_{∞} (cm)	k (a^{-1})	accroissement mensuel (cm)
Pacifique Ouest (1)	65,5	0,94	1,9
Pacifique Est (1)	79	0,64	2
Pacifique Ouest (2)	61,3	1,25	2
Pacifique Est (2)	75,5	0,77	2,2
Pacifique Central (3)	102	0,55	2,7
Océan Indien (4)	60,6	0,93	1,5
Atlantique Est tropical (5)	80	0,60	2
Présente étude (hypothèse 1)	60	0,75	1,1
Présente étude (hypothèse 2)	62	1,1	1,9

- (1) JOSSE et al (1979)
- (2) SIBERT et al (1983)
- (3) UCHIYAMA et STRUHSACKER (1981)
- (4) MARCILLE et STEQUERT (1976)
- (5) BARD et ANTOINE (1983)

C O N C L U S I O N

La méthode de PAULY modifiée a permis d'évaluer les vitesses de croissance des thons jaunes et des bonites pêchés à la traîne à Vanuatu en 1983. On a constaté que celle des thons jaunes était lente relativement à ce qui a pu être observé dans d'autres zones pour des individus de taille supérieure à 70 cm. Les valeurs obtenues pour les bonites sont du même ordre de grandeur que celles trouvées par d'autres auteurs. Les difficultés inhérentes à l'analyse des distributions de fréquences tiennent à l'existence de plus d'un recrutement annuel. Les résultats obtenus seront vérifiés par l'étude des stries sur épines dorsales. Par ailleurs, le système permanent de collecte et de saisie des données qui a été mis en place à Vanuatu, permettra d'ajuster les résultats année après année.

R E M E R C I E M E N T S

Les auteurs tiennent à remercier le Service des Pêches, et en particulier son Directeur, Monsieur CROSSLAND, pour le souci constant d'étroite collaboration dont il fait preuve.

B I B L I O G R A P H I E

- ALBARET, J.J. - 1977 - La reproduction de l'albacore (Thunnus albacares) dans le Golfe de Guinée, Cah. ORSTOM, sér. océanogr., 15(4) : 389-419
- BARD, F.X. et L. ANTOINE - 1983 - Croissance du listao dans l'Atlantique Est. Document provisoire présenté à la Conférence ICCAT de l'année internationale du listao. Ténériffe, juin 1983, 24 p.
- BAYLIFF, W.H. - 1973 - Observations on the growth of yellowfin tuna in the eastern Pacific Ocean derived from tagging experiments. Inter Amer. Trop. Tuna Comm., Int. Rep. 7 : 1,26.
- CHABANNE, J., L. MAREC et A. ASINE - 1982 - La pêche bonitière en Polynésie Française en 1980, ORSTOM Tahiti, Notes et documents d'Océanographie, 15, 43 p.
- FONTENEAU, A. - 1979 - Croissance de l'albacore (Thunnus albacares) de l'Atlantique Est. ICCAT, Recueil de Documents Scientifiques vol. IX, n° 1 (SCRS/79/51) : 152-169.
- GRANDPERRIN, R. et F. BROUARD - 1983 - Etat d'avancement des recherches conduites par l'ORSTOM en matière de pêche à Vanuatu. Mission ORSTOM de Port-Vila, Notes et documents d'Océanographie, 6, 19 p.
- JOSSE, E., J.C. LE GUEN, R. KEARNEY, A. LEWIS, A. SMITH, L. MAREC et P.K. TOMLINSON - 1979 - Croissance des bonites à ventre rayé. Commission du Pacifique Sud, Nouméa, Doc. occasionnel 11, 83 p.
- LAUREC A., et J.C. LE GUEN - 1981 - Dynamique des populations marines exploitées. Tome 1, concepts et modèles. Centre National pour l'Exploitation des Océans. Rapports scientifiques et techniques 45, 118 p.
- LE GUEN, J.C., et G.T. SAKAGAWA - 1973 - Apparent growth of yellowfin tuna from the Eastern Atlantic Ocean. Fishery Bull., 71(1) : 175-187.
- MARCILLE, J. et B. STEQUERT - 1976 - Etude préliminaire de la croissance du listao (Katsuwonus pelamis) dans l'ouest de l'Océan Indien tropical. Cah. ORSTOM, sér. Océanogr., 14(2) : 139-151.

- MARCILLE, J. et B. STEQUERT - 1976 - Croissance des jeunes albacores Thunnus albacares et patudos Thunnus obesus de la côte nord-ouest de Madagascar. Cah. ORSTOM, sér. Océanogr., 14(2) : 153-162.
- PAULY, D., et N. DAVID - 1981 - ELEFAN I, a BASIC program for the objective extractions of growth parameters from length - frequency data. Meeresforsch. 28(4) : 205-211.
- PAULY, D., N. DAVID et J. INGLES - 1980 - ELEFAN I : User's instruction and program listings (rev. 2). Mimeo. pag. var.
- PRESTON, G. - 1982 - The fijian experience in the utilisation of fish aggregating devices. Commission du Pacifique Sud, SPC/Fisheries 14/WP 25, 61 p.
- SHOMURA, R.S. et W.M. MATSUMOTO - 1982 - Structured flotsam as fish aggregating devices. Southwest Fish. Cent., Nat. Mar. Fish. Serv., NOAA, Honolulu, NOAA - TM - NMFS - SWFC - 22, 9 p.
- SIBERT, J.R., R.E. KEARNEY et T.A. LAWSON - 1983 - Variations in growth increments of tagged skipjack (Katsuwonus pelamis). South Pacific Commission, Tuna and Billfish Assessment Programme Tech. Rep. 10, 43 p.
- STEQUERT, B. - 1976 - Etude de la maturité sexuelle de la ponte et de la fécondité du listao (Katsuwonus pelamis) de la côte nord-ouest de Madagascar. Cah. ORSTOM, sér. Océanogr., 14(3) : 227-247.
- SUZUKI, Z. - 1971 - Comparison of growth parameters estimated for the yellowfin tuna in the Pacific Ocean. Bull. Far Seas Fish. Res. Lab. 5 : 89-105
- UCHIYAMA, J.H. et P. STRUHSAKER - 1981 - Age and growth of skipjack tuna, Katsuwonus pelamis, and yellowfin tuna, Thunnus albacares, as indicated by daily growth increments of sagittae. Fishery Bull., 79(1) : 151-162.

Annexe 1 - Logiciel d'analyse des distributions de fréquences de tailles
dérivé du programme ELEFAN 1 de PAULY et DAVID (1981)

a) Saisie des données

```
10 ! PROGRAMME P1/SAI
20 ! SAISIE DES DONNEES POUR TR
   AITEMENT ELEFAN 1
30 !
40 !
50 OPTION BASE 0
60 DIM N$(39)
70 CLEAR
80 DISP "SAISIE DES DONNEES POU
   R ANALON" @ DISP
90 DISP "Nom du fichier ";@ INP
   UT F$
100 CLEAR
110 DISP "Nom de l'espece ";@ IN
   PUT N$
120 DISP "Nbre d'echantillons ";
   @ INPUT N1
130 DISP "Taille d'un intervalle
   ";@ INPUT A3
140 DISP "Taille minimale ";@ IN
   PUT A1
150 DISP "Taille maximale ";@ IN
   PUT A2
160 N2=(A2-A1)/A3
170 DISP "Le fichier est-il cree
   : O/N";@ INPUT R0$
180 IF R0$="O" THEN 220
190 IF R0$#"N" THEN 170
200 N=100+40*N1*N2
210 CREATE F$,1,N
220 ASSIGN# 1 TO F$
230 PRINT# 1 ; N$,N1,N2,A3,A1,A2
240 FOR S1=0 TO N1-1
250 CLEAR @ L,F=0 @ I=1
260 DISP "Mois et jour de l'echa
   ntillon";S1+1 @ INPUT M,D
270 CLEAR @ DISP "Echantillon";S
   1+1 @ DISP "Mois";M
280 IF I=2 THEN DISP "Donnees pr
   ecedentes";L-A3;F
290 ON I GOSUB 370,390
300 IF F#9999 THEN PRINT# 1 ; S1
   ,L,F,M,D @ L=L+A3 @ GOTO 270
310 NEXT S1
320 S1=9999
330 PRINT# 1 ; S1,L,F,M,D
340 ASSIGN# 1 TO *
350 CLEAR @ DISP "FIN DE SAISIE
   DES DONNEES ."
360 END
370 DISP @ DISP "Longueur, freu
   ence ";@ INPUT L,F
380 I=2 @ RETURN
390 DISP @ DISP "Longueur ";L
400 DISP "Frequence (9999 si
   'FIN') ";@ INPUT F
410 RETURN
```

b) Analyse des données

```
10 REM ELEFAN I VERSION 2
20 ! MODIFICATIONS URSTOM VANUA
   TU DU 01/08/84
30 ! PROGRAMME P2/ANA
40 !
50 OPTION BASE 0
60 CLEAR
70 DISP "          ELEFAN I VERS
   2          " @ DISP
80 U2=0
90 COM T0(12),SHORT S(30,30)
100 COM F1(12,60),F0(60),L1(48),
   L2(48),N$(39)
110 BEEP @ DISP "Nom du fichier"
;
120 INPUT F$
130 BEEP @ DISP "Installez la K7
   ; puis faire CONT" @ PAUSE
140 ASSIGN# 1 TO F$
150 BEEP @ CLEAR @ DISP TAB(6);"
   INITIALISATION..."
160 FOR I=0 TO 11
170 T0(I)=0
180 FOR J=0 TO 60 @ F1(I,J)=0 @
   NEXT J
190 NEXT I
200 BEEP @ CLEAR @ DISP "IMPRESS
   ION DES DONNEES: O/N";@ INPU
   T S$
210 IF S$#"O" AND S$#"N" THEN 20
   @
220 READ# 1 ; N$,N1,N2,A3,A1,A2
230 IF S$="N" THEN 310
240 PRINT @ PRINT "ESPECE: ";N$
   @ PRINT
250 PRINT F$ @ PRINT "====="
260 PRINT "NB ECHANT.";N1
270 PRINT "NB INTERV.";N2
280 PRINT "TAILLE INTERVALLE";A3
290 PRINT "LONG. MIN.";A1
300 PRINT "LONG. MAX.";A2
310 READ# 1 ; S1,L,F,M,D
320 FOR I=0 TO N1-1
330 IF S$="N" THEN 360
340 PRINT @ PRINT "ECHANTILLON";
   S1+1 @ PRINT D;"/";M
350 PRINT @ PRINT "LONGUEUR ";TA
   B(15);"FREQUENCE" @ PRINT
360 L1(I)=L
370 T0(I)=M+(D-1)/31
380 IF S1>I THEN 470
390 J=(L-L1(I))/A3+1
400 F1(I,J)=F
410 L3=L
420 IF S$="N" THEN 440
430 PRINT L;TAB(15);F
440 READ# 1 ; S1,L,F,M,D
450 IF S1>9999 THEN 470
460 GOTO 380
470 L2(I)=L3
480 J1=1
490 J2=(L2(I)-L1(I))/A3+1
500 R=0
510 FOR J=J1 TO J2
520 K1=J-1
530 K2=J-2
540 W1=J+1
550 W2=J+2
560 F0(J)=0
570 IF K2<0 THEN K2=0
580 IF W1>J2 THEN W1,W2=0 @ GOTO
   600
590 IF W2>J2 THEN W2=0
600 M2=F1(I,J)+F1(I,K1)+F1(I,K2)
   +F1(I,W1)+F1(I,W2)
610 M2=M2/5
620 IF M2#0 THEN F0(J)=F1(I,J)/M
   2 @ R=R+F0(J)
630 NEXT J
640 R=R/J2
650 IF R=0 THEN 690
660 FOR J=J1 TO J2
670 F1(I,J)=F0(J)/R-1
680 NEXT J
690 FOR J=J1 TO J2
700 IF F1(I,J)<0 THEN GOTO 850
710 K1=J-1
720 K2=J-2
730 W1=J+1
740 W2=J+2
750 IF W1>J2 THEN W1,W2=0 @ GOTO
   770
760 IF W2>J2 THEN W2=0
770 IF K2<0 THEN K2=0
780 M2=0
790 F1(I,0)=-1
800 IF F1(I,K1)<=-1 THEN M2=M2+1
810 IF F1(I,K2)<=-1 THEN M2=M2+1
820 IF F1(I,W1)<=-1 THEN M2=M2+1
830 IF F1(I,W2)<=-1 THEN M2=M2+1
840 F1(I,J)=F1(I,J)*EXP(-M2)
850 NEXT J
860 M2=0
870 M3=0
880 FOR J=J1 TO J2
890 IF F1(I,J)>0 THEN M2=M2+F1(I
   ,J) @ GOTO 910
900 M3=M3+F1(I,J)
910 NEXT J
920 R=1
930 IF M3<>0 THEN R=M2/M3
940 FOR J=J1 TO J2
950 IF F1(I,J)<0 AND F1(I,J)<>-1
   THEN F1(I,J)=F1(I,J)*(-R)
960 NEXT J
970 REM END OF DEEMPHASIZE
980 REM COMPUTE ASP
990 J=J1
1000 IF J>J2 THEN GOTO 1100
1010 IF F1(I,J)<=0 THEN J=J+1 @
   GOTO 1000
```

```

1020 U1=F1(I,J)
1030 J3=J+1
1040 IF J3>J2 THEN U2=U2+U1 @ GO
TO 1100
1050 IF F1(I,J3)>U1 THEN U1=F1(I
,J3) @ J3=J3+1 @ GOTO 1040
1060 IF F1(I,J3)>0 THEN J3=J3+1
@ GOTO 1040
1070 U2=U2+U1
1080 J=J3+1
1090 GOTO 1000
1100 FOR J=1 TO J2
1110 IF F1(I,J)=-1 THEN F1(I,J)=
0
1120 NEXT J
1130 NEXT I
1140 CLEAR
1150 ASSIGN# 1 TO *
1160 BEEP @ DISP "IMPRESSION DE
L'ECHANTILLON RE- STRUCTURE
: O/N "; @ INPUT S$
1170 IF S$="O" AND S$#"N" THEN 1
160
1180 IF S$="N" THEN 1280
1190 PRINT @ PRINT "ECHANTILLON
RESTRUCTURE "
1200 PRINT "-----
-----"
1210 FOR I=0 TO N1-1 @ PRINT @ P
RINT "ECHANTILLON";I+1 @ PR
INT
1220 PRINT "LONGUEUR";TAB(15);"F
REQUENCE" @ PRINT
1230 FOR J=1 TO (L2(I)-L1(I))/A3
+1
1240 PRINT L1(I)+(J-1)*A3;TAB(15
);F1(I,J)
1250 NEXT J
1260 NEXT I
1270 PRINT @ PRINT "ASP=";U2
1280 CLEAR @ FOR I=0 TO 9 @ BEEP
@ NEXT I
1290 DISP " <<< SURFACE DES ES
P >>> " @ DISP
1300 DISP "Longevite "; @ INPUT T
9
1310 DISP "Linf minimum, Linf ma
ximum "; @ INPUT Z0,Z9
1320 DISP "Pas de Linf "; @ INPUT
Z1
1330 DISP "K minimum, K maximum
"; @ INPUT K0,K9
1340 DISP "Pas de K "; @ INPUT K1
1350 DISP "Echantillon de depart
"; @ INPUT C0 @ C0=C0-1
1360 IF C0>N1-1 THEN BEEP @ GOTO
1350
1370 DISP "Longueur initiale "; @
INPUT L0
1380 DISP "Increment de longueur
initiale "; @ INPUT P1
1390 IF P1=0 THEN P1=A3
1400 I9=(Z9-Z0)/Z1 @ J9=(K9-K0)/
K1
1410 Z=Z0-Z1
1420 FOR I=0 TO I9
1430 Z=Z+Z1 @ K=K0-K1
1440 FOR J=0 TO J9
1450 K=K+K1
1460 S0,R=0 @ P=L0
1470 CLEAR @ DISP USING "K,2D,A,
2D,A" ; "CALCULS DE S(",I,"
",J,")"
1480 T=FNB2(P) @ S(I,J),R1=0 @ C
1=C0
1490 C2=C1 @ GOSUB 1800
1500 C2=(C1+1) MOD N1
1510 T1=(T0(C2)-T0(C1)) MOD 12
1520 IF T1=0 THEN T1=12
1530 T1=T1/12 @ T=T+T1
1540 IF T>T9 THEN 1590
1550 GOSUB 1800
1560 IF R1=1 THEN 1590
1570 C1=C2
1580 GOTO 1500
1590 T=FNB2(P) @ R1=0 @ C1=C0
1600 C2=(C1-1) MOD N1
1610 T1=(T0(C1)-T0(C2)) MOD 12
1620 IF T1=0 THEN T1=12
1630 T1=T1/12 @ T=T-T1
1640 IF T<0 THEN 1690
1650 GOSUB 1800
1660 IF R1=1 THEN 1690
1670 C1=C2
1680 GOTO 1600
1690 IF S(I,J)>S0 THEN S0=S(I,J)
@ P2=P
1700 P=P+P1 @ IF P<L0+A3 THEN 14
80
1710 S(I,J)=S0
1720 NEXT J
1730 NEXT I
1740 CLEAR @ DISP "Enregistremen
t: O/N "; @ INPUT R0$
1750 IF R0$="N" THEN 2020
1760 IF R0$#"O" THEN 1740
1770 DISP "Nom du fichier"; @ INP
UT F$
1780 DISP "Le fichier est-il cre
e: O/N "; @ INPUT R0$
1790 IF R0$="O" THEN 1810
1800 IF R0$#"N" THEN 1780 ELSE N
=120+(I9+1)*(J9+1)*8 @ CREA
TE F$,I,N
1810 ASSIGN# 1 TO F$
1820 PRINT# 1 ; N$,C0+1,L0,Z0,Z9
,Z1,K0,K9,K1
1830 FOR I=0 TO I9 @ FOR J=0 TO
J9 @ PRINT# 1 ; S(I,J) @ NE
XT J @ NEXT I
1840 ASSIGN# 1 TO *
1850 CLEAR @ DISP "ENREGISTREMEN
T TERMINE"
1860 STOP
1870 !
1880 !
1890 L=FNB1(T)
1900 IF (L<A2)*(L>A1)=0 THEN R1=
1 @ RETURN
1910 C=(A3+L-L1(C2))\A3
1920 IF C<1 THEN RETURN
1930 S(I,J)=S(I,J)+F1(C2,C)
1940 RETURN
1950 DEF FNB1(X)
1960 X=EXP(-(K*X))
1970 FNB1=Z*(1-X)
1980 FN END
1990 DEF FNB2(X)
2000 FNB2=LOG(Z/(Z-X))/K
2010 FN END
2020 END

```

c) Représentation de la surface des ESP

```
10 ! ORSTOM VANUATU
20 ! PROGRAMME P3/ESP
30 ! EDITION SURFACE DES ESP
40 !
50 !
60 PLOTTER IS 705
70 DEG
80 OPTION BASE 0
90 DIM N$(40)
100 COM S(30,30)
110 CLEAR @ DISP "          SORTIE DES
    SURFACES ESP " @ DISP
120 DISP "Nom du fichier ",@ INP
    UT F$
130 CLEAR @ DISP "Installez la k
    7 adequat; puis faire CONT
    " @ PAUSE
140 CLEAR @ ASSIGN# 1 TO F$
150 READ# 1 ; N$,E0,L,L0,L9,L1,K
    0,K9,K1
160 I9=(L9-L0)/L1
170 J9=(K9-K0)/K1
180 FOR I=0 TO I9
190 FOR J=0 TO J9
200 READ# 1 ; S(I,J)
210 IF I=0 AND J=0 THEN M0,M9=S(
    0,0) @ GOTO 240
220 IF S(I,J)>M9 THEN M9=S(I,J)
230 IF S(I,J)<M0 THEN M0=S(I,J)
240 NEXT J
250 NEXT I
260 !
270 !
280 !
290 LIMIT 10,250,10,180
300 LOCATE 10,10+70*J9/I9,20,90
310 SCALE K0-K1,K9+K1,L0-L1,L9+L
    1
320 FXD 2,1 @ PEN 1
330 CSIZE 1.8,.6
340 LAXES -K1,L1,K0-K1,L0-L1,4,4
350 CSIZE 1.8,.6,15
360 MOVE K0-K1,L9+L1*1.5 @ LORG
    4 @ LABEL "Linf (cm)"
370 MOVE K9+K1,L0-2*L1 @ LORG 9
    @ LABEL "k (1/a)"
380 LORG 6 @ MOVE K9+2*K1,L0+(L9
    -L0)/2 @ LDIR 90
390 LABEL USING "19A,22" ; "Echa
    ntillon init.:",E0
400 LABEL USING "16A,30.0" ; "Lo
    ngueur init.:",L
410 LABEL USING "8A,20.40,4X,8A,
    20.40" ; "ESPmin: ",M0,"ESPm
    ax: ",M9
420 LDIR 0
430 CSIZE 2.5,.9,15 @ LORG 4 @ M
    OVE K0+(K9-K0)/2,L9+3*L1
440 LABEL "SURFACE DES ESP" @ LA
    BEL N$
450 SCALE 0,J9+2,0,I9+2
460 PEN 2
470 FOR I=0 TO I9
480 FOR J=0 TO J9
490 C=(S(I,J)-M0)/(M9-M0)
500 C=IP(C*10)+1
510 GOSUB 550
520 NEXT J
530 NEXT I
540 STOP
550 !
560 MOVE J+.5,I+.5
570 K=0
580 IMOVE 0,1/C @ K=K+1/C
590 IF K>1 THEN RETURN ELSE IDRA
    W 1,0
600 IMOVE 0,1/C @ K=K+1/C
610 IF K>1 THEN RETURN ELSE IDRA
    W -1,0
620 GOTO 580
```

d) Représentation des courbes de croissance

```

10 ! PROGRAMME P4/CRB
20 ! REPRESENTATION DES COURBES
  DE CROISSANCE
30 !
40 !
50 OPTION BASE 1
60 INTEGER C1
70 COM T(12,50),M(12),L1(12),L2
  (12),N#[40]
80 S0=INF
90 CLEAR @ DISP "Nom du fichier
  " ; @ INPUT F$
100 DISP "Faire CONT si pret." @
  PAUSE
110 CLEAR @ DISP "LECTURE DU FIC
  HIER"
120 ASSIGN# 1 TO F$
130 READ# 1 ; N$,N1,N2,A3,A1,A2
140 READ# 1 ; S1,L,F,M,D
150 IF S1=9999 THEN 210
160 IF S0#S1 THEN I=S1+1 @ S0=S1
  @ L1(I)=L
170 L2(I)=L
180 T(I,1+(L-L1(I))\A3)=F
190 M(I)=M+(D-1)/31
200 GOTO 140
210 CLEAR @ DISP "CALCULS"
220 T9=0
230 FOR I=1 TO N1
240 T0=0 @ J1=1 @ J2=1+(L2(I)-L1
  (I))\A3
250 FOR J=J1 TO J2 @ T0=T0+T(I,J
  ) @ NEXT J
260 FOR J=J1 TO J2
270 T(I,J)=T(I,J)*100/T0
280 IF T(I,J)>T9 THEN T9=T(I,J)
290 NEXT J
300 NEXT I
310 GOSUB 740
320 CLEAR @ DISP "Histogramme (1
  ) ou courbe (2) " ; @ INPUT C8
330 DISP "Echelle maximum des fr
  equences " ; @ INPUT C7
340 C=0 @ F9=C7*.95/T9
350 FOR I=1 TO N1
360 IF N1=1 THEN P=C+1+M(I)/12 E
  LSE P=M(I)+C
370 J2=1+(L2(I)-L1(I))\A3
380 IF C8=1 THEN 400
390 MOVE P,L1 @ DRAW P,L2
400 FOR J=J1 TO J2
410 F=F9*T(I,J) @ L=A3*(J-1)+L1(
  I)
420 IF J=J1 THEN MOVE P-F,L+A3/2
430 IF C8=1 THEN CLIP P-F,P,L,L+
  A3 @ FRAME ELSE DRAW P-F,L+A
  3/2
440 NEXT J
450 NEXT I
460 IF N1#1 THEN 490
470 C=C+1
480 IF C=C0-1 THEN 510 ELSE 350
490 C=C+12
500 IF C+1<=C0*12 THEN 350
510 !
520 !
530 !
540 CLEAR @ DISP " PARAMETRES DE
  VAN BERTALANFFY" @ DISP
550 DISP "Entrez Linf,K " ; @ INPU
  T Z,K
560 DISP "Longueur et echantillo
  n de depart " ; @ INPUT X
  0,00
570 DISP "Temps minimal (en anne
  e) " ; @ INPUT S0
580 DISP "Type de ligne " ; @ INPU
  T V0
590 CLIP C1,C2,L1,L2
600 IF N1=1 THEN S9=S0+C2 ELSE S
  9=1+(S0+C0)*12
610 S2=(M(D0)-1)/12 @ S1=FNB2(X0
  )
620 S2=FP(S1)-S2
630 LINETYPE V0,3
640 IF N1=1 THEN GOSUB 1000 ELSE
  GOSUB 1150
650 CLEAR @ DISP "Legende parame
  tres: O/N " ; @ INPUT R0$
660 IF R0$="O" THEN 680
670 IF R0$="N" THEN 540 ELSE BEE
  P @ GOTO 650
680 DISP "Longueur de l'echantil
  lon " ; @ INPUT E0
690 DISP "Position " ; @ INPUT E1,
  E2
700 UNCLIP
710 T$=VAL$(Z)&"/"&VAL$(K)
720 MOVE E1,E2 @ IDRAW E0,0 @ IM
  OVE E0*.2,0 @ LOG2 @ CSIZE
  2,1,.15 @ LABEL T$
730 GOTO 540
740 !
750 !
760 !
770 CLEAR @ DISP "S'assurer que
  le PLOTTER est surON; puis f
  aire CONT." @ PAUSE
780 PLOTTER IS 705
790 DEG
800 CLEAR @ DISP "Lmin=";A1;"Lma
  x=";A2
810 DISP "Echelle des longueurs
  (min,max) " ; @ INPUT L1,L2
820 DISP "Grad. laire, frequence
  des grad.2aires " ; @ INPUT L
  3,L4
830 DISP "Longevite " ; @ INPUT C0
840 DISP "Etiquette annuelle (1)
  ou men- suelle (2) " ; @ INP
  UT C9

```

```
850 IF C9=2 AND N1=1 THEN BEEP @
    GOTO 840
860 CLEAR
870 IF N1=1 THEN C2=C0 @ C1=0 EL
    SE C2=1+C0*12 @ C1=-1
880 LIMIT 0,250,0,180
890 LOCATE 10,130,30,90
900 CSIZE 2,1
910 SCALE 0,C2,L1,L2
920 LAXES 0,L3,0,0,0,L4
930 MOVE 0,L2*1.02 @ LORG 4 @ CS
    IZE 2,1,15 @ LABEL "Longueur
    (cm)"
940 SCALE C1,C2,L1,L2
950 XAXIS L1,1,1,C2
960 LORG 6 @ CSIZE 2,1
970 FOR I=1 TO C2
980 MOVE I,L1-(L2-L1)*.01
990 ON C9 GOSUB 1040,1070
1000 NEXT I
1010 MOVE C2,L1-(L2-L1)*.05 @ LO
    RG 9 @ CSIZE 2,1,15
1020 IF N1=1 THEN LABEL "Temps (
    annee)" ELSE LABEL "Temps (
    mois)"
1030 RETURN
1040 IF N1=1 THEN LABEL I @ RETU
    RN
1050 IF (I-1)\12=(I-1)/12 THEN L
    ABEL (I-1)/12
1060 RETURN
1070 LABEL 1+(I-1) MOD 12 @ RETU
    RN
1080 !
1090 MOVE 0,FNB1(S0)
1100 FOR S=S0 TO S9 STEP .1
1110 DRAW S-S0,FNB1(S+S2)
1120 NEXT S
1130 PENUP
1140 RETURN
1150 !
1160 S0=S0*12
1170 MOVE 1,FNB3(S0)
1180 FOR S=S0 TO S9 STEP .2
1190 DRAW S-S0+1,FNB3(S)
1200 NEXT S
1210 PENUP
1220 RETURN
1230 !
1240 DEF FNB1(X)
1250 X=EXP(-(K*X))
1260 FNB1=Z*(1-X)
1270 FN END
1280 DEF FNB2(X)
1290 FNB2=LOG(Z/(Z-X))/K
1300 FN END
1310 DEF FNB3(X)
1320 M=X MOD 12/12
1330 A=X\12
1340 X1=A+M+S2
1350 FNB3=FNB1(X1)
1360 FN END
```

Annexe 2 - Restructuration des échantillons (Thunnus albacares, Vanuatu)

ESPECE: THUNNUS albacares

THALB

=====

NB ECHANT. 7

NB INTERV. 15

TAILLE INTERVALLE 2

LONG. MIN. 28

LONG. MAX. 58

ECHANTILLON 1

1 / 5

LONGUEUR	FREQUENCE	FREQUENCE ECHANTILLON RESTRUCTURE
30	14	-.595172950826
32	82	.226973325864
34	116	.64970604932
36	165	1.36365127484
38	38	-.318437016759
40	11	-.683510006225
42	3	-.643210676201

ECHANTILLON 2

1 / 6

LONGUEUR

30	1	-.981303853497
32	57	-.294799324334
34	300	2.22305960578
36	183	.91683778088
38	82	-.061804997034
40	17	-.603537464753
42	3	-.806926333231
44	2	-.39152541382

ECHANTILLON 3

1 / 7

LONGUEUR

28	2	.539893809012
30	1	-.708082340483
32	3	-.775607895914
34	22	-.363217862291
36	106	1.67320229546
38	80	1.03552465894
40	15	-.518286120777
42	1	-.838379652244
44	3	-.128983177104
46	1	.083936285406

ECHANTILLON 4

1 / 8

LONGUEUR

28	1	.072551568092
30	0	0
32	4	- .447325369609
34	15	- .161933107308
36	25	.14474701573
38	63	1.86257199903
40	22	.10401169548
42	5	- 63277256839

ECHANTILLON 5

1 / 9

LONGUEUR

34	2	- .855384374186
36	28	- .110394786702
38	57	.42291446106
40	111	1.62509980905
42	54	.39795104946
44	16	- .470250128619
46	5	- .606826267333
48	4	-3.21877349847E-2
50	1	- .370922027737

ECHANTILLON 6

1 / 10

LONGUEUR

36	6	- .834774619423
38	61	.121432462424
40	158	1.81910661255
42	84	.467354287
44	18	- .62042228284
46	13	- .388725406047
48	9	.22117026211
50	1	- .757869582633
52	2	-2.72717331385E-2

ECHANTILLON 7

1 / 11

LONGUEUR

38	13	- .698766286441
40	190	1.31406993337
42	142	.890545162
44	58	- .200176425312
46	18	- .503916894241
48	12	- .277865390419
50	4	- .412976634083
52	6	.34521607865
54	2	- .121967801575
56	1	- 334161741945